**Exercise 02 (Assignment 04) :**

```
0x01800513         addi a0, zero, 24   # A = 24
0x00300593         addi a1, zero, 3    # B = 3
0x00000393         addi t2, zero, 0    # i = 0
0x00058E33         add  t3, a1, zero   # temp = B = 3
0x01C54863   loop: blt  a0, t3, done   # if A < temp, done
0x00138393         addi t2, t2, 1      # i++
0x00BE0E33         add  t3, t3, a1     # temp += B
0xFF5FF06F         j    loop           # repeat loop
0x00038533   done: add a0, t2, zero    # result = i
```

(b) An equivalent program in C would be (assuming A = a0, B = a1, i = t2, temp = t3, and result = a0 once the assembly program reaches the done label:

```
int A = 24;
int B = 3;
int i;
int result;

int temp = B;
for (i = 0; A >= temp; i++)
  temp += B;

result = i;
```

(c) This program performs integer division: result = A/B.

**Exercise 03 (Assignment 04) :**

```
0x01F00393       addi t2,   zero, 31    # t2 = 31
0x00755E33   L1: srl  t3,   a0,   t2    # t3 = a0 >> t2
0x001E7E13       andi t3,   t3,   1     # t3 = lsb of t3
0x01C580A3       sb   t3,   1(a1)       # a1[1] = t3
0x00158593       addi a1,   a1,   1     # a1++
0xFFF38393       addi t2,   t2,  -1     # t2--
0xFE03D6E3       bge  t2,   zero, L1    # if t2 >= 0, repeat
0x00008067       jr   ra                # return
```

```
# a0 = val, a1 = base address of array
# t2 = shiftAmt, t3 = tmp
void decToBin(int val, char array[]){
  int shiftAmt = 31;
  char tmp;
  int i = 1;

  do {
    tmp = (val >> shiftAmt);
    tmp = tmp & 1;
    array[i] = tmp;
    i++;
    shiftAmt--;
  } while (shiftAmt >= 0);
}
```

**Exercise 06 (Assignment 04) :**

| Instruction | (a) Machine Code | (b) Type | Addressing mode |
|---|---|---|---|
| addi t4, a1, 0 | 00058E93 | I | Immediate |
| ori a0, a0, 32 | 02056513 | I | Immediate |
| sub a1, a1, a0 | 40A585B3 | R | Register-only |
| jal Func2 | 024000EF | J | PC-relative |
| lw t2, 4(a0) | 00452383 | I | Base |
| sw t2, 16(a1) | 0075A823 | S | Base |
| srli t3, t2, 8 | 0083DE13 | I | Immediate |
| beq t2, t3, Else | 01C38463 | B | PC-relative |
| jr ra | 00008067 | I | Immediate |
| addi a0, a0, 4 | 00450513 | I | Immediate |
| j Func2 | FE9FF06F | J | PC-relative |